MAPLE PROCEDURES FOR ANALYZING THE ACCURACY OF THE APPROXIMATE DEVELOPMENTS OBTAINED USING GORE METHOD

Mădălina Roxana BUNECI, University Constantin Brâncuşi of Târgu-Jiu, ROMÂNIA

Abstract: *The purpose of this paper is provide Maple procedures to quantify how close are the original 3D-object and the 3D-object formed using an approximate development obtained by applying the well-known gore method to a surface of revolution.*

Keywords: pattern development; approximate model; surfaces of revolution; distortion; volume; metrics.

1. INTRODUCTION

Developable surfaces are surfaces that can be isometrically mapped to a plane. Therefore developable surfaces can be unfolded (developed) into a plane without introducing any distortion. Industries where 3D objects are manufacturing with materials that do not stretch or tear commonly employ this kind of surfaces. The modeling techniques for complicated free form shapes may lead to non-developable surfaces, which should be converted into a set of developable pieces. In this paper we consider approximate developments of a surface of revolution obtained as in [2] (using the gore method) and we provide Maple procedures to quantify how close are the 3D-object obtained by rolling or folding back the approximate development (2D-pattern) and the original 3D-object. The motivation of this paper is to facilitate studies as those in [4-5], for instance. By contrast to [6-8] where the distortion introduced when parameterizing a surface into the plane is taken into consideration, here we take into account the volumes of the 3D-objects whose boundaries are the original surface (1.3) and the surface obtained by rolling or folding back the approximate development of (1.3).

We start with a possibly non-developable surface of revolution and its associated approximate development constructed as in [2]. Specifically, we take into consideration surface of revolutions obtained by rotating a plane curve around the z-axis (vertical axis). We assume that the curve is defined by its parametric form

$$\begin{cases} x=x(t) \\ y=0, \quad t\in[a, b] \\ z=z(t) \end{cases} \qquad (1.1)$$

and that the following conditions are satisfied:

$$\begin{cases} 1.\ \ x(t) \geq 0 \text{ for all } t\in[a, b]. \\ 2.\ \ t\to x(t) \text{ and } t\to z(t) \text{ are piecewise continuously differentiable.} \\ 3.\ \ z'(t)\neq 0 \text{ for all but finitely many } t \in[a,b]. \end{cases} \qquad (1.2)$$

The standard parameterization of the surface of revolution obtained by rotating the plane curve (1.1) about the z-axis is

$$\begin{cases} x = x(t)\cos(\theta) \\ y = x(t)\sin(\theta), \quad t\in [a, b],\ \theta\in,\ 2\pi] \end{cases} \qquad (1.3)$$

z = z(t)

In the case of a non-complete rotation, we may assume $\theta\in[\theta_1, \theta_2]$, and in this case the results of the procedure in Section 3 should be multiply by $(\theta_2- \theta_1)/(2\pi)$.

2. APPROXIMATE DEVELOPMENTS AND THEIR ASSOCIATED 3D-OBJECTS

We consider a set of parameters $\Delta=\{t_1, t_2, \ldots, t_m\}\subset[a.b]$ be such that

$$a=t_1<t_2<\ldots<t_{m-1}<t_m=b.$$

As in [2], the surface of revolution (1.3) is divided into a number of n equal sections of $2\pi/n$ intervals by vertical planes and in m divisions obtained by the intersections with the horizontal planes: $z=z(t_1)$, $z= z(t_2)$, …, $z=z(t_m)$.



To each section $j\in\{1,2,\ldots,n\}$ we associate an arch shaped sector (gore) $D_1D_2\ldots D_mE_mE_{m-1}\ldots E_1$ constructed as follows: for every $i\in\{1,2,\ldots,m\}$, let

$$B_i\left(x(t_i)\cos\left(\frac{2\pi}{n}(j-1)\right),x(t_i)\sin\left(\frac{2\pi}{n}(j-1)\right),z(t_i)\right),C_i\left(x(t_i)\cos\left(\frac{2\pi}{n}j\right),x(t_i)\sin\left(\frac{2\pi}{n}j\right),z(t_i)\right)$$

the intersections of the surface (1.3) and the plane $z=z(t_i)$ and the vertical planes that define the section. We also take into consideration the point

$$A_i\left(x(t_i)\cos\left(\frac{\pi(2j-1)}{n}\right),x(t_i)\sin\left(\frac{\pi(2j-1)}{n}\right),z(t_i)\right),$$

that cuts the arc B_iC_i (contained in the intersection of the surface (1.3) and the plane $z=z(t_i)$) in two arcs of equal lengths. We map A_i to the point A'_i $(X_i,0)$ into a plane P (with a fixed coordinate system (X,Y)) such that $|A'_1A'_i|$ = length of the curve γ_i specified parametrically by

$$\begin{cases} x= x(t)\cos(\pi(2j-1)/n) \\ y= x(t)\sin(\pi(2j-1)/n) \\ z=z(t) \end{cases} \qquad t\in[t_1, t_i]$$

The line $A'_1A'_m$ will be the symmetry axis for $D_1D_2\ldots D_mE_mE_{m-1}\ldots E_1$. We map into the same plane the points B_i, respectively C_i to $D_i(X_i,Y_i)$, respectively, $E_i(X_i, -Y_i)$ such that $2|A_iD_i|=2|A_iE_i|=|D_iE_i|$ = length of the arc B_iC_i (contained in the intersection of the surface (1.3) and the plane $z=z(t_i)$). Hence $|D_iE_i| = 2\pi x(t_i)/n$ and $Y_i=\pi x(t_i)/n$. The arch shaped sector (gore) is obtained joining the points D_1, D_2, …, D_m (and symmetrically, E_1, E_2, …, E_m) by a smooth curve. The curve joining the points D_1, D_2, …, D_m can be seen as a smooth function f with the property that $f(A'_i)=D_i$ for all $i\in\{1,2,\ldots,m\}$. As we have noted in [2], if we use polynomial interpolation to construct f, the Runge phenomenon could appear. To avoid this problem we use spline curves of degree d.

Let us denote by O_0 the original 3D-object and by O_a the 3D-object obtained by sticking/gluing together these n gores (congruent to $D_1D_2\ldots D_mE_mE_{m-1}\ldots E_1$). In order to quantify how close O_a and O_0 are, we approximate O_a by a polyhedron whose vertices are determined as follows. We consider a refinement Δ' of Δ (this means $\Delta\subset\Delta'$). Let $\Delta'=\{s_1, s_2, \ldots, s_{m'}\}\subset[a, b]$ and let's assume that

$$a=s_1<s_2<\ldots<s_{m'-1}<s_{m'}=b.$$

The curves joining the points D_1, D_2, …, D_m, respectively, E_1, E_2, …, E_m of a gore are approximated by the polylines (polygonal curves) corresponding to the refinement Δ'. Furthermore, the 3D-object O_a is approximated by a polyhedron obtained by gluing the corresponding line segments of the polylines associated to consecutive gores. This polyhedron is a union of regular pyramid frustums and regular prisms.



If the sampling Δ' is fine enough, the resulted polyhedron will be a good approximation of O_a.

The next procedure plots a partial approximation of O_a corresponding to the refinement Δ' using n_0 of the n arch shaped sectors (gores)

```
> draw_gore_app_3D := proc(x, z, parameter, refine_param, theta1,
theta2, n, d, n0)
local mr, pointsxr, pointsyr, i, j, p, r, h, theta;
 theta := (theta2 - theta1)/n;
 mr := ArrayNumElems(refine_param);
 pointsxr := Array(1 .. mr);
 pointsxr[1] := 0;
 for i to mr - 1 do pointsxr[i + 1] := pointsxr[i] +
  curve_length(x, z, refine_param[i],refine_param[i + 1])
 end do;
 pointsyr := Array(1 .. mr);
 for i to mr do pointsyr[i] := evalf(eval(approximation_gore(x, z,
             parameter, theta1, theta2, n,d, X), X = pointsxr[i]))
 end do;
 p := Array(1 .. n0);
 r := Array(1 .. mr);
```

```
 h := Array(1 .. mr);
 for i to mr do
  r[i] := pointsyr[i]/sin(1/2*theta);
  h[i] := z(refine_param[i])
 end do;
 for j to n0 do p[j] := seq(plottools:-polygon([
   [r[i]*cos(theta1 + (j-1)*theta),r[i]*sin(theta1 + (j-1)*theta),
     h[i]],
   [r[i]*cos(theta1 + j*theta),r[i]*sin(theta1 + j*theta), h[i]],
   [r[i + 1]*cos(theta1 + j*theta),r[i + 1]*sin(theta1 + j*theta),
     h[i + 1]],
   [r[i+1]*cos(theta1 + (j-1)*theta),
      r[i+1]*sin(theta1 + (j-1)*theta), h[i + 1]]
   ]),  i = 1 .. mr - 1)
 end do;
 plots:-display(seq(p[j], j = 1 .. n0), scaling = constrained,
   axes = none)
end proc
```

The procedures curve_length, and approximation_gore used in the above procedure where defined in [2].

$n=8$, $n_0=8$



$m=4$, $m'=4$ $m=4$, $m'=7$ $m=4$, $m'=16$ $m=4$, $m'=49$ $m=8$, $m'=8$ $m=8$, $m'=29$

3. MAPLE PROCEDURES FOR ANALYZING THE ACCURACY OF THE APPROXIMATE DEVELOPMENTS OBTAINED USING GORE METHOD

We consider the standard parameterization (1.3) of a general surface of revolution, a set of parameters $\Delta=\{t_0, t_1, \ldots, t_m\}\subset[a,b]$ and **a** refinement $\Delta' =\{s_1, s_2, \ldots, s_{m'}\}$ of Δ that satisfy the below conditions (3.1)

1) $a=s_1<s_2<\ldots<s_{m'-1}<s_{m'}=b$.
2) $t\to x(t)$ and $t\to z(t)$ are continuously differentiable on (s_i, s_{i+1}) for every $i\in\{1,\ldots,m'\}$
3) $z(t)\neq 0$ for all $t \in[a,b]\setminus \{s_0, s_1, \ldots, s_{m'}\}$

In order to quantify how close the 3D-object O_a (formed using the approximate development) is from the original 3D-object O_0, we compute the volume V of $(O_a\setminus O_0) \cup (O_0\setminus O_a)$ ("approximation error"). In order to compute V, we approximate O_a with the polyhedron P(Δ') associated to Δ' (described in Section 2). Then $V \approx \sum_{i=1}^{m'} \mathrm{vol}(S_i(\Delta'))$, where for every $i \in \{1,\ldots,m'\}$,

$$\mathrm{vol}(S_i(\Delta')) = \iiint_{S_i(\Delta')} 1 = \int_{z(s_i)}^{z(s_{i+1})} A_h dh,$$

and where A_h is the area between the intersection curves of the plane z=h and the surface (1.3), respectively, the boundary of the polyhedron **P(Δ')**. By (3.1), $z|[s_i, s_{i+1}]$ is strictly increasing or decreasing, hence injective. Thus $\mathrm{vol}(S_i(\Delta')) = \int_{s_i}^{s_{i+1}} A_{z(t)} z'(t) dt$.

For instance, for n=8 and $t \in [a,b]$, $A_{z(t)}$ could be the area of one of the following regions colored in red:



In all procedures provide in this paper the integrals are computed using Maple evalf(Int(..)), i.e. using numerical integration (see, for instance [1] for an introduction in numerical integration). The reason is to avoid the case when the symbolic int fails and returns an unevaluated integral.

The next procedure computes $V \approx \sum_{i=1}^{m'} \mathrm{vol}(S_i(\Delta'))$

```
>error_approximation_gore_vol_b := proc(x, z, parameter,
          refine_param, n, d)
   local mr, pointsxr, pointsyr, i, j, r, theta, er_approx, d10,
         vol, f, a, af, rt, ct;
   d10 := Digits;
   Digits := d10 + 3;
   theta := 2*Pi/n;
   mr := ArrayNumElems(refine_param);
   pointsxr := Array(1 .. mr);
   pointsxr[1] := 0;
   for i to mr - 1 do
    pointsxr[i+1] := pointsxr[i] +
      curve_length(x, z, refine_param[i],refine_param[i + 1])
   end do;
   pointsyr := Array(1 .. mr);
   f := unapply(
        approximation_gore(x, z, parameter, 0, 2*Pi, n, d, X)
        , X);
   for i to mr do pointsyr[i] := evalf(f(pointsxr[i]))
   end do;
   r := Array(1 .. mr);
   er_approx := 0;
   vol := 0;
   for i to mr do r[i] := evalf(pointsyr[i]/sin(1/2*theta))
```

```
    end do;
    for i to mr - 1 do
     rt := abs(evalf(r[i + 1] +
             (z(refine_param[i + 1]) - z(t))*(r[i] - r[i + 1])
             /(z(refine_param[i + 1]) - z(refine_param[i]))));
     ct := rt*cos(1/2*theta)/abs(x(t));
     a := piecewise(
      0 <= abs(x(t))-rt or evalf(abs(x(t))- rt*cos(1/2*theta)) < 0,
      abs(Pi*x(t)^2 - 1/2*n*rt^2*sin(theta)),
       abs(1/2*n*rt^2*sin(theta) - Pi*x(t)^2 +
       n*x(t)^2*(evalf(arccos(ct))-sin(1/2*evalf(2*arccos(ct))))));
     af := unapply(a, t);
     er_approx := er_approx + evalf(
     Int(af, refine_param[i] .. refine_param[i + 1]));
     vol := vol + Pi*evalf(Int(x(t)^2*diff(z(t), t),
                        t = refine_param[i] .. refine_param[i + 1]))
   end do;
   Digits := d10;
   print(`Absolute Error`, evalf(er_approx));
   print(`Relative Error (%)`, evalf(100*er_approx/vol))
end proc
```

Applying this procedure to [Example 2.6, 3], it results a smaller error than in the case of zone methods treated in [3].

Also if we use spline functions of degree d to approximate the gores, let V(d) denotes the error returned by the above procedure. For the [Example 2.6, 3] we have V(2)<V(1)<V(3), while for the case of the sphere, we have V(3)<V(2)<V(1).

REFERENCES

[1] M. Buneci, Metode Numerice – aspecte teoretice şi practice, Editura Academica Brâncuşi, 2009.

[2] M. Buneci, Approximate developments for surfaces of revolution, Fiabilitate și durabilitate (Fiability & durability), 2 (2016), 143-149

[3] M. Buneci, Maple procedures for error estimation of approximate developments obtained using horizontal zones, Annals of the "Constantin Brâncuși" University of Târgu-Jiu. Engineering Series, 4 (2017), 134-139.

[4] L. Luca, The development of a rotation surface generated by hypocycloid with 4 branches, Durabilitate si fiabilitate, 2 (2016), 64-67.

[5] L. Luca, A study of approximate development of spherical surfaces, June 2019 JIDEG (Journal of Industrial Design and Engineering Graphics), to appear.

[6] P. V. Sander, J. Snyder, S. J. Gortler, and H. Hoppe, Texture mapping progressive meshes, In SIGGRAPH '01: Proceedings of the 28th annual conference on Computer graphics and interactive techniques, pp. 409–416. ACM Press, 2001.

[7] P. V. Sander, S. J. Gortler, J. Snyder, and H. Hoppe, Signal-specialized parametrization, In EGRW '02: Proceedings of the 13th Eurographics workshop on Rendering, pp. 87–98. Eurographics Association, 2002.

[8] A. Sheffer, B. Lévy, M. Mogilnitsky and A. Bogomyakov, ABF++ : Fast and Robust Angle Based Flattening, ACM Transactions on Graphics, Association for Computing Machinery, 2004, 24 (2), 311-330.